Markov Chains

Basics

```
MarkovStep[s_, K_] := RandomChoice[K[[s]] -> Range[Dimensions[K][[2]]]]
MarkovChain[s_, K_, t_] := NestList[MarkovStep[#, K] &, s, t]   (* A sample of t
  steps from a Markov chain with transition matrix K starting from state s *)

TransitionsOf[K_] := Flatten[Table[If[K[[i, j]] > 0, i ð j, {}],
   {i, Dimensions[K][[1]]}, {j, Dimensions[K][[2]]}]]
 (* The edges of the directed graph of probable transitions
  associated to the transition matrix K *)
```

Examples

$$\text{RWC}_5 = \begin{pmatrix} 0 & 0.5 & 0 & 0 & 0.5 \\ 0.5 & 0 & 0.5 & 0 & 0 \\ 0 & 0.5 & 0 & 0.5 & 0 \\ 0 & 0 & 0.5 & 0 & 0.5 \\ 0.5 & 0 & 0 & 0.5 & 0 \end{pmatrix};$$

```
(* Simple random walk on C5 *)
```

$$\text{K}_1 = \begin{pmatrix} 0.4 & 0.3 & 0 & 0 & 0.3 \\ 0.5 & 0 & 0.4 & 0.1 & 0.0 \\ 0 & 0.5 & 0 & 0.5 & 0 \\ 0.1 & 0 & 0.5 & 0 & 0.4 \\ 0.4 & 0.1 & 0 & 0.5 & 0 \end{pmatrix};$$

Visualization

Active State

```
ActiveStateStyle[x_] := {_ → Gray, x → Yellow}
MarkovActivePlot[K_, x_] :=
 Graph[TransitionsOf[K], VertexStyle → ActiveStateStyle[x],
  VertexSize → Medium, VertexLabels → Placed["Name", Center]]
```

Distribution

```
DistributionStateStyle[u_] := Table[i → u[[i]], {i, 1, Length[u]}]
MarkovDistributionPlot[K_, u_] :=
 Graph[TransitionsOf[K], VertexStyle → RGBColor[0.25, 0.25, 1],
  VertexSize → DistributionStateStyle[u], VertexLabels → "Name"]
```

Example

```
u = {0.1, 0.2, 0.3, 0.15, 0.25};
```

```
{MarkovActivePlot[RWC5, 2], MarkovDistributionPlot[RWC5, u]}
```




```
{MarkovActivePlot[K1, 2], MarkovDistributionPlot[K1, u]}
```



Evolution

Evolution of the distribution

```
u = {0.05, 0.2, 0.35, 0.15, 0.25};
```

Random walk on C_5

K_1

```
Manipulate[MarkovDistributionPlot[K1, u.MatrixPower[K1, t]],
 {t, 0, 100, 1, Appearance → "Open", AnimationRate → 4}]
```



Evolution of the state

Random walk on C_5

K_5

```
X0 = 3;  (* Initial state *)
T = 100;  (* Number of time steps *)
X = MarkovChain[X0, K1, T];
Manipulate[MarkovActivePlot[K1, X[[t + 1]]],
 {t, 0, T - 1, 1, Appearance → "Open", AnimationRate → 5}]
```



Evolution of empirical distribution

Random walk on C_5

K_5

```
u0 = {1, 0, 0, 0, 0};  (* Distribution of the initial state *)
T = 100;  (* Number of time steps *)
n = 1000;  (* Number of independent samples *)
XS0 = RandomChoice[u0 → Range[Length[u0]], n];  (* n independent samples from u0 *)
XS = Table[MarkovChain[XS0[[k]], K1, T], {k, 1, n}];
h = Table[Normalize[
    BinCounts[Transpose[XS][[t + 1]], {0.5, Length[u0] + 0.5, 1}], Total], {t, 0, T - 1}];
(* Empirical distribution of n independent runs in T consecutive steps *)
Manipulate[MarkovDistributionPlot[K1, h[[t + 1]]],
 {t, 0, T - 1, 1, Appearance → "Open", AnimationRate → 4}]
```